


Computation Structures

Procedures & Stacks Worksheet



PUSH(X): Push $\text{Reg}[x]$ onto stack

```
ADDC(SP, 4, SP)
```

```
ST(Rx, -4, SP)
```

POP(X): Pop value at top of stack into $\text{Reg}[x]$

```
LD(SP, -4, RX)
```

```
SUBC(SP, 4, SP)
```

ALLOCATE(k): Reserve k words of stack


```
ADDC(SP, 4*k, SP)
```

DEALLOCATE(k): Release k words of stack

```
SUBC(SP, 4*k, SP)
```

Stack discipline: leave stack the way you found it => for every **PUSH()**, there's a corresponding **POP()** or **DEALLOCATE()**

Activation record layout on the stack (aka stack frame):



CALLING SEQUENCE

```
PUSH(argn) // push args, last arg first
```

```
...
```

```
PUSH(arg1)
```

```
BR(f, LP) // call f, return addr in LP
```

```
DEALLOCATE(n) // remove args from stack
```

ENTRY SEQUENCE

```
f: PUSH(LP) // save return addr
```

```
PUSH(BP) // save old frame pointer
```

```
MOVE(SP, BP) // initialize new frame pointer
```

```
ALLOCATE(nlocals) // make room for locals
```

```
(push other regs) // preserve old reg vals
```

EXIT SEQUENCE

```
// return value in R0
```

```
MOVE(BP, SP) // remove locals
```

```
POP(BP) // restore old frame pointer
```

```
POP(LP) // recover return address
```

```
JMP(LP) // resume execution in caller
```

Problem 1.

You are given an incomplete listing of a C program (shown below) and its translation to Beta assembly code (shown on the right):

```
int fn(int x) {
    int lowbit = x & 1;
    int rest = x >> 1;
    if (x == 0) return 0;
    else return ???;
}
```

```
fn: PUSH(LP)
    PUSH(BP)
    MOVE(SP, BP)
    ALLOCATE(2)
    PUSH(R1)
    LD(BP, -12, R0)
    ANDC(R0, 1, R1)
xx: ST(R1, 0, BP)
    SHRC(R0, 1, R1)
    ST(R1, 4, BP)
yy: BEQ(R0, rtn)
    LD(BP, 4, R1)
    PUSH(R1)
    BR(fn, LP)
    DEALLOCATE(1)
    LD(BP, 0, R1)
    ADD(R1, R0, R0)
rtn: POP(R1)
zz: MOVE(BP, SP)
    POP(BP)
    POP(LP)
    JMP(LP)
```

- (A) What is the missing C source corresponding to ??? in the above program?

C source code: _____

- (B) Suppose the instruction bearing the tag 'zz:' were eliminated from the assembly language program. Would the modified procedure work the same as the original procedure (circle one)?

Work the same? YES ... NO

- (C) In the space below, fill in the binary representation for the instruction stored at the location tagged 'xx:' in the above program.

--	--	--	--

(fill in missing 1s and 0s for instruction at xx:)

The procedure **fn** is called from an external procedure and its execution is interrupted just prior to the execution of the instruction tagged '**yy**'. The contents of a region of memory are shown on the left below.

NB: All addresses and data values are shown in hex. The contents of **BP** are 0x1C8 and **SP** contains 0x1D4.

184:	4	
188:	7	
18C:	47	
190:	C4	
194:	170	
198:	1	
19C:	23	
1A0:	22	
1A4:	23	
1A8:	4C	
1AC:	198	
1B0:	1	
1B4:	11	
1B8:	23	
1BC:	11	
1C0:	4C	
1C4:	1B0	
1C8:	1	←BP
1CC:	8	
1D0:	???	
1D4:	0	←SP

(D) What was the argument to the most recent call to **fn**?

Most recent argument (HEX): x= _____

(E) What is the missing value marked ??? for the contents of location 1D0?

Contents of 1D0 (HEX): _____

(F) What is the hex address of the instruction tagged **rtn**:?

Address of rtn (HEX): _____

(G) What was the argument to the *original* call to **fn**?

Original argument (HEX): x= _____

(H) What is the hex address of the BR instruction that called **fn** *originally*?

Address of original call (HEX): _____

(I) What were the contents of R1 at the time of the *original* call?

Original R1 contents (HEX): _____

(J) What value will be returned to the *original* caller?

Return value for original call (HEX): _____

Problem 2.

You are given an incomplete listing of a C program (shown below) and its translation to Beta assembly code (shown on the right):

```
int f(int x, int y) {
    x = (x >> 1) + y;
    if (y == 0) return x;
    else return ???;
}
```

```
f:  PUSH(LP)
    PUSH(BP)
    MOVE(SP, BP)
    PUSH(R1)
    LD(BP, -12, R0)
    SHRC(R0, 1, R0)
    LD(BP, -16, R1)
    ADD(R0, R1, R0)
    BEQ(R1, rtn)
    SUBC(R1, 1, R1)
    PUSH(R1)
    PUSH(R0)
    BR(f, LP)
DEALLOCATE(2)
rtn: POP(R1)
zz:  MOVE(BP, SP)
    POP(BP)
    POP(LP)
    JMP(LP)
```

(A) What is the missing C source corresponding to ??? in the above program?

C source code: _____

(B) Suppose the instruction bearing the tag ‘zz:’ were eliminated from the assembly language program. Would the modified procedure work the same as the original procedure?

Work the same (circle one)? YES ... NO

The procedure **f** is called from an external procedure and then execution is stopped just prior to one of the executions of the instruction labeled ‘**rtn:**’. The addresses and contents of a region of memory are shown in the table on the right; all addresses and data values in the table are in hex. When execution is stopped **BP** contains the value **0x14C** and **SP** contains the value **0x150**.

108	7
10C	320
110	104
114	3
118	A
11C	2C4
120	104
124	3
128	2
12C	
130	348
134	124
138	2
13C	1
140	6
144	348
148	138
14C	1
150	0
154	4
158	348
15C	14C
160	0

(C) What are the arguments to the **currently active call** to **f**?

Most recent arguments (in hex): x = 0x_____, y = 0x_____

(D) If you can tell from the information provided, specify the arguments to the **original** call to **f**, otherwise select **CAN’T TELL**.

Original arguments (in hex) : x = 0x_____, y = 0x_____, or **CAN’T TELL**

(E) What is the missing value in location 0x12C?

Contents of location 0x12C (in hex): 0x_____

(F) What is the hex address of the instruction labeled **rtn:**?

Address of instruction labeled rtn: (in hex): 0x_____

(G) What is the hex address of the BR instruction that called **f** *originally*?

Address of original call (in hex): 0x_____, or **CAN’T TELL**

(H) What value will be returned to the *original* caller?

Return value for original call (in hex): 0x_____

MIT OpenCourseWare
<https://ocw.mit.edu/>

6.004 Computation Structures
Spring 2017

For information about citing these materials or our Terms of Use, visit: <https://ocw.mit.edu/terms>.